
Description

arima fits univariate models for a time series, where the disturbances are allowed to follow a
linear autoregressive moving-average (ARMA) specification. When independent variables are included
in the specification, such models are often called ARMAX models; and when independent variables are
not specified, they reduce to Box–Jenkins autoregressive integrated moving-average (ARIMA) models
in the dependent variable.

Quick start
AR(1) model using tsset data

arima y, ar(1)

MA(1) model
arima y, ma(1)

ARMA(2,1) model
arima y, ar(1/2) ma(1)

Same as above
arima y, arima(2,0,1)

Same as above, and take first difference of y and restrict estimation to years 1990 to 2010
arima D.y if tin(1990,2010), ar(1/2) ma(1)

Same as above
arima y if tin(1990,2010), arima(2,1,1)

Multiplicative SARIMA model with quarterly data
arima y, arima(2,1,1) sarima(2,1,0,4)

ARMAX model with covariates x1 and x2, an AR(1) process, and robust standard errors
arima y x, ar(1) vce(robust)

Menu
Statistics > Time series > ARIMA and ARMAX > ARIMA and ARMAX models

Syntax

Basic syntax for a regression model with ARMA disturbances

arima depvar
[

indepvars
]
, ar(numlist) ma(numlist)

Basic syntax for an ARIMA(p, d, q) model

arima depvar , arima(#p,#d,#q)

Basic syntax for a multiplicative seasonal ARIMA(p, d, q)× (P,D,Q)s model

arima depvar , arima(#p,#d,#q) sarima(#P ,#D,#Q,#s)

Full syntax

arima depvar
[

indepvars
] [

if
] [

in
] [

weight
] [

, options
]

options Description

Model

noconstant suppress constant term
arima(#p,#d,#q) specify ARIMA(p, d, q) model for dependent variable
ar(numlist) autoregressive terms of the structural model disturbance
ma(numlist) moving-average terms of the structural model disturbance
constraints(constraints) apply specified linear constraints

Model 2

sarima(#P ,#D,#Q,#s) specify period-#s multiplicative seasonal ARIMA term
mar(numlist, #s) multiplicative seasonal autoregressive term; may be repeated
mma(numlist, #s) multiplicative seasonal moving-average term; may be repeated

Model 3

condition use conditional MLE instead of full MLE
savespace conserve memory during estimation
diffuse use diffuse prior for starting Kalman filter recursions
p0(# |matname) use alternate prior for starting Kalman recursions; seldom used
state0(# |matname) use alternate state vector for starting Kalman filter recursions

SE/Robust

vce(vcetype) vcetype may be opg, robust, or oim

Reporting

level(#) set confidence level; default is level(95)

detail report list of gaps in time series
nocnsreport do not display constraints
display options control columns and column formats, row spacing, and line width

Maximization

maximize options control the maximization process; seldom used

collinear keep collinear variables
coeflegend display legend instead of statistics

https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.1.3ifexp
https://www.stata.com/manuals/u11.pdf#u11.1.4inrange
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/r.pdf#rvce_option

You must tsset your data before using arima; see [TS] tsset.
depvar and indepvars may contain time-series operators; see [U] 11.4.4 Time-series varlists.
by, collect, fp, rolling, statsby, and xi are allowed; see [U] 11.1.10 Prefix commands.
iweights are allowed; see [U] 11.1.6 weight.
collinear and coeflegend do not appear in the dialog box.
See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Options

� � �
Model �

noconstant; see [R] Estimation options.

arima(#p,#d,#q) is an alternative, shorthand notation for specifying models with ARMA disturbances.
The dependent variable and any independent variables are differenced #d times, and 1 through #p
lags of autocorrelations and 1 through #q lags of moving averages are included in the model. For
example, the specification

. arima D.y, ar(1/2) ma(1/3)

is equivalent to

. arima y, arima(2,1,3)

The latter is easier to write for simple ARMAX and ARIMA models, but if gaps in the AR or MA
lags are to be modeled, or if different operators are to be applied to independent variables, the
first syntax is required.

ar(numlist) specifies the autoregressive terms of the structural model disturbance to be included in
the model. For example, ar(1/3) specifies that lags of 1, 2, and 3 of the structural disturbance
be included in the model; ar(1 4) specifies that lags 1 and 4 be included, perhaps to account for
additive quarterly effects.

If the model does not contain regressors, these terms can also be considered autoregressive terms
for the dependent variable.

ma(numlist) specifies the moving-average terms to be included in the model. These are the terms for
the lagged innovations (white-noise disturbances).

constraints(constraints); see [R] Estimation options.

If constraints are placed between structural model parameters and ARMA terms, the first few
iterations may attempt steps into nonstationary areas. This process can be ignored if the final
solution is well within the bounds of stationary solutions.

� � �
Model 2 �

sarima(#P ,#D,#Q,#s) is an alternative, shorthand notation for specifying the multiplicative seasonal
components of models with ARMA disturbances. The dependent variable and any independent
variables are lag-#s seasonally differenced #D times, and 1 through #P seasonal lags of autoregressive
terms and 1 through #Q seasonal lags of moving-average terms are included in the model. For
example, the specification

. arima DS12.y, ar(1/2) ma(1/3) mar(1/2,12) mma(1/2,12)

is equivalent to

. arima y, arima(2,1,3) sarima(2,1,2,12)

https://www.stata.com/manuals/tstsset.pdf#tstsset
https://www.stata.com/manuals/u11.pdf#u11.4.4Time-seriesvarlists
https://www.stata.com/manuals/u11.pdf#u11.1.10Prefixcommands
https://www.stata.com/manuals/u11.pdf#u11.1.6weight
https://www.stata.com/manuals/u20.pdf#u20Estimationandpostestimationcommands
https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions

mar(numlist, #s) specifies the lag-#s multiplicative seasonal autoregressive terms. For example,
mar(1/2,12) requests that the first two lag-12 multiplicative seasonal autoregressive terms be
included in the model.

mma(numlist, #s) specifies the lag-#s multiplicative seasonal moving-average terms. For example,
mma(1 3,12) requests that the first and third (but not the second) lag-12 multiplicative seasonal
moving-average terms be included in the model.

� � �
Model 3 �

condition specifies that conditional, rather than full, maximum likelihood estimates be produced.
The presample values for εt and µt are taken to be their expected value of zero, and the estimate
of the variance of εt is taken to be constant over the entire sample; see Hamilton (1994, 132).
This estimation method is not appropriate for nonstationary series but may be preferable for long
series or for models that have one or more long AR or MA lags. diffuse, p0(), and state0()
have no meaning for models fit from the conditional likelihood and may not be specified with
condition.

If the series is long and stationary and the underlying data-generating process does not have a long
memory, estimates will be similar, whether estimated by unconditional maximum likelihood (the
default), conditional maximum likelihood (condition), or maximum likelihood from a diffuse
prior (diffuse).

In small samples, however, results of conditional and unconditional maximum likelihood may
differ substantially; see Ansley and Newbold (1980). Whereas the default unconditional maximum
likelihood estimates make the most use of sample information when all the assumptions of the model
are met, Harvey (1989) and Ansley and Kohn (1985) argue for diffuse priors often, particularly in
ARIMA models corresponding to an underlying structural model.

The condition or diffuse options may also be preferred when the model contains one or more
long AR or MA lags; this avoids inverting potentially large matrices (see diffuse below).

When condition is specified, estimation is performed by the arch command (see [TS] arch),
and more control of the estimation process can be obtained using arch directly.

condition cannot be specified if the model contains any multiplicative seasonal terms.

savespace specifies that memory use be conserved by retaining only those variables required for
estimation. The original dataset is restored after estimation. This option is rarely used and should
be used only if there is not enough space to fit a model without the option. However, arima
requires considerably more temporary storage during estimation than most estimation commands
in Stata.

diffuse specifies that a diffuse prior (see Harvey 1989 or 1993) be used as a starting point for the
Kalman filter recursions. Using diffuse, nonstationary models may be fit with arima (see the
p0() option below; diffuse is equivalent to specifying p0(1e9)).

By default, arima uses the unconditional expected value of the state vector ξt (see Methods and
formulas) and the mean squared error (MSE) of the state vector to initialize the filter. When the
process is stationary, this corresponds to the expected value and expected variance of a random draw
from the state vector and produces unconditional maximum likelihood estimates of the parameters.
When the process is not stationary, however, this default is not appropriate, and the unconditional
MSE cannot be computed. For a nonstationary process, another starting point must be used for the
recursions.

In the absence of nonsample or presample information, diffuse may be specified to start the
recursions from a state vector of zero and a state MSE matrix corresponding to an effectively

https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/u11.pdf#u11.1.8numlist
https://www.stata.com/manuals/tsarch.pdf#tsarch

infinite variance on this initial state. This method amounts to an uninformative and improper prior
that is updated to a proper MSE as data from the sample become available; see Harvey (1989).

Nonstationary models may also correspond to models with infinite variance given a particular
specification. This and other problems with nonstationary series make convergence difficult and
sometimes impossible.

diffuse can also be useful if a model contains one or more long AR or MA lags. Computation
of the unconditional MSE of the state vector (see Methods and formulas) requires construction
and inversion of a square matrix that is of dimension {max(p, q + 1)}2, where p and q are the
maximum AR and MA lags, respectively. If q = 27, for example, we would require a 784-by-784
matrix. Estimation with diffuse does not require this matrix.

For large samples, there is little difference between using the default starting point and the diffuse
starting point. Unless the series has a long memory, the initial conditions affect the likelihood of
only the first few observations.

p0(# |matname) is a rarely specified option that can be used for nonstationary series or when an
alternate prior for starting the Kalman recursions is desired (see diffuse above for a discussion
of the default starting point and Methods and formulas for background).

matname specifies a matrix to be used as the MSE of the state vector for starting the Kalman filter
recursions—P1|0. Instead, one number, #, may be supplied, and the MSE of the initial state vector
P1|0 will have this number on its diagonal and all off-diagonal values set to zero.

This option may be used with nonstationary series to specify a larger or smaller diagonal for P1|0
than that supplied by diffuse. It may also be used with state0() when you believe that you
have a better prior for the initial state vector and its MSE.

state0(# |matname) is a rarely used option that specifies an alternate initial state vector, ξ1|0 (see
Methods and formulas), for starting the Kalman filter recursions. If # is specified, all elements of
the vector are taken to be #. The default initial state vector is state0(0).

� � �
SE/Robust �

vce(vcetype) specifies the type of standard error reported, which includes types that are robust to
some kinds of misspecification (robust) and that are derived from asymptotic theory (oim, opg);
see [R] vce option.

For state-space models in general and ARMAX and ARIMA models in particular, the robust or
quasimaximum likelihood estimates (QMLEs) of variance are robust to symmetric nonnormality in
the disturbances, including, as a special case, heteroskedasticity. The robust variance estimates are
not generally robust to functional misspecification of the structural or ARMA components of the
model; see Hamilton (1994, 389) for a brief discussion.

� � �
Reporting �

level(#); see [R] Estimation options.

detail specifies that a detailed list of any gaps in the series be reported, including gaps due to
missing observations or missing data for the dependent variable or independent variables.

nocnsreport; see [R] Estimation options.

display options: noci, nopvalues, vsquish, cformat(% fmt), pformat(% fmt), sformat(% fmt),
and nolstretch; see [R] Estimation options.

https://www.stata.com/manuals/rvce_option.pdf#rvce_option
https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions
https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions
https://www.stata.com/manuals/d.pdf#dformat
https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions

� � �
Maximization �

maximize options: difficult, technique(algorithm spec), iterate(#),
[
no
]
log, trace,

gradient, showstep, hessian, showtolerance, tolerance(#), ltolerance(#),
nrtolerance(#), gtolerance(#), nonrtolerance(#), and from(init specs); see [R] Maxi-
mize for all options except gtolerance(), and see below for information on gtolerance().

These options are sometimes more important for ARIMA models than most maximum likelihood
models because of potential convergence problems with ARIMA models, particularly if the specified
model and the sample data imply a nonstationary model.

Several alternate optimization methods, such as Berndt–Hall–Hall–Hausman (BHHH) and Broyden–
Fletcher–Goldfarb–Shanno (BFGS), are provided for ARIMA models. Although ARIMA models are
not as difficult to optimize as ARCH models, their likelihoods are nevertheless generally not quadratic
and often pose optimization difficulties; this is particularly true if a model is nonstationary or
nearly nonstationary. Because each method approaches optimization differently, some problems
can be successfully optimized by an alternate method when one method fails.

Setting technique() to something other than the default or BHHH changes the vcetype to vce(oim).

The following options are all related to maximization and are either particularly important in fitting
ARIMA models or not available for most other estimators.

technique(algorithm spec) specifies the optimization technique to use to maximize the
likelihood function.

technique(bhhh) specifies the Berndt–Hall–Hall–Hausman (BHHH) algorithm.

technique(dfp) specifies the Davidon–Fletcher–Powell (DFP) algorithm.

technique(bfgs) specifies the Broyden–Fletcher–Goldfarb–Shanno (BFGS) algorithm.

technique(nr) specifies Stata’s modified Newton–Raphson (NR) algorithm.

You can specify multiple optimization methods. For example, technique(bhhh 10 nr 20)
requests that the optimizer perform 10 BHHH iterations, switch to Newton–Raphson for 20
iterations, switch back to BHHH for 10 more iterations, and so on.

The default is technique(bhhh 5 bfgs 10).

gtolerance(#) specifies the tolerance for the gradient relative to the coefficients. When
|gi bi| ≤ gtolerance() for all parameters bi and the corresponding elements of the
gradient gi, the gradient tolerance criterion is met. The default gradient tolerance for arima
is gtolerance(.05).

gtolerance(999) may be specified to disable the gradient criterion. If the optimizer
becomes stuck with repeated “(backed up)” messages, the gradient probably still contains
substantial values, but an uphill direction cannot be found for the likelihood. With this option,
results can often be obtained, but whether the global maximum likelihood has been found
is unclear.

When the maximization is not going well, it is also possible to set the maximum number
of iterations (see [R] Maximize) to the point where the optimizer appears to be stuck and
to inspect the estimation results at that point.

from(init specs) allows you to set the starting values of the model coefficients; see [R] Max-
imize for a general discussion and syntax options.

The standard syntax for from() accepts a matrix, a list of values, or coefficient name value
pairs; see [R] Maximize. arima also accepts from(armab0), which sets the starting value
for all ARMA parameters in the model to zero prior to optimization.

https://www.stata.com/manuals/rmaximize.pdf#rMaximizeSyntaxalgorithm_spec
https://www.stata.com/manuals/rmaximize.pdf#rMaximize
https://www.stata.com/manuals/rmaximize.pdf#rMaximize
https://www.stata.com/manuals/rmaximize.pdf#rMaximize
https://www.stata.com/manuals/rmaximize.pdf#rMaximize
https://www.stata.com/manuals/rmaximize.pdf#rMaximize
https://www.stata.com/manuals/rmaximize.pdf#rMaximize

ARIMA models may be sensitive to initial conditions and may have coefficient values that
correspond to local maximums. The default starting values for arima are generally good,
particularly in large samples for stationary series.

The following options are available with arima but are not shown in the dialog box:

collinear, coeflegend; see [R] Estimation options.

Remarks and examples stata.com

Remarks are presented under the following headings:
Introduction
ARIMA models
Multiplicative seasonal ARIMA models
ARMAX models
Dynamic forecasting
Video example

Introduction
arima fits both standard ARIMA models that are autoregressive in the dependent variable and

structural models with ARMA disturbances. Good introductions to the former models can be found in
Box et al. (2016); Hamilton (1994); Harvey (1993); Newton (1988); Diggle (1990); and many others.
The latter models are developed fully in Hamilton (1994) and Harvey (1989), both of which provide
extensive treatment of the Kalman filter (Kalman 1960) and the state-space form used by arima to fit
the models. Becketti (2020) discusses ARIMA models and Stata’s arima command, and he devotes an
entire chapter explaining how the principles of ARIMA models are applied to real datasets in practice.

Consider a first-order autoregressive moving-average process. Then arima estimates all the pa-
rameters in the model

yt = xtβ + µt structural equation
µt = ρµt−1 + θεt−1 + εt disturbance, ARMA(1, 1)

where
ρ is the first-order autocorrelation parameter
θ is the first-order moving-average parameter
εt ∼ i.i.d. N(0, σ2), meaning that εt is a white-noise disturbance

You can combine the two equations and write a general ARMA(p, q) in the disturbances process as

yt = xtβ + ρ1(yt−1 − xt−1β) + ρ2(yt−2 − xt−2β) + · · ·+ ρp(yt−p − xt−pβ)

+ θ1εt−1 + θ2εt−2 + · · ·+ θqεt−q + εt

It is also common to write the general form of the ARMA model more succinctly using lag operator
notation as

ρ(Lp)(yt − xtβ) = θ(Lq)εt ARMA(p, q)

where
ρ(Lp) = 1− ρ1L− ρ2L2 − · · · − ρpLp

θ(Lq) = 1 + θ1L+ θ2L
2 + · · ·+ θqL

q

and Ljyt = yt−j .

Multiplicative seasonal ARMAX and ARIMA models can also be fit.

https://www.stata.com/manuals/restimationoptions.pdf#rEstimationoptions
http://stata.com

For stationary series, full or unconditional maximum likelihood estimates are obtained via the
Kalman filter. For nonstationary series, if some prior information is available, you can specify initial
values for the filter by using state0() and p0() as suggested by Hamilton (1994) or assume an
uninformative prior by using the diffuse option as suggested by Harvey (1989).

Missing data are allowed and are handled using the Kalman filter and methods suggested by Harvey
(1989 and 1993); see Methods and formulas.

ARIMA models

Pure ARIMA models without a structural component do not have regressors and are often written
as autoregressions in the dependent variable, rather than autoregressions in the disturbances from a
structural equation. For example, an ARMA(1, 1) model can be written as

yt = α+ ρyt−1 + θεt−1 + εt (1a)

Other than a scale factor for the constant term α, these models are equivalent to the ARMA in the
disturbances formulation estimated by arima, though the latter are more flexible and allow a wider
class of models.

To see this effect, replace xtβ in the structural equation above with a constant term β0 so that

yt = β0 + µt

= β0 + ρµt−1 + θεt−1 + εt

= β0 + ρ(yt−1 − β0) + θεt−1 + εt

= (1− ρ)β0 + ρyt−1 + θεt−1 + εt (1b)

Equations (1a) and (1b) are equivalent, with α = (1−ρ)β0, so whether we consider an ARIMA model
as autoregressive in the dependent variable or disturbances is immaterial. Our illustration can easily
be extended from the ARMA(1, 1) case to the general ARIMA(p, d, q) case.

Example 1: ARIMA model

Enders (2004, 87–93) considers an ARIMA model of the U.S. Wholesale Price Index (WPI)
using quarterly data over the period 1960q1 through 1990q4. The simplest ARIMA model that includes
differencing and both autoregressive and moving-average components is the ARIMA(1,1,1) specification.
We can fit this model with arima by typing

. use https://www.stata-press.com/data/r18/wpi1

. arima wpi, arima(1,1,1)

(setting optimization to BHHH)
Iteration 0: Log likelihood = -139.80133
Iteration 1: Log likelihood = -135.6278
Iteration 2: Log likelihood = -135.41838
Iteration 3: Log likelihood = -135.36691
Iteration 4: Log likelihood = -135.35892
(switching optimization to BFGS)
Iteration 5: Log likelihood = -135.35471
Iteration 6: Log likelihood = -135.35135
Iteration 7: Log likelihood = -135.35132
Iteration 8: Log likelihood = -135.35131

ARIMA regression

Sample: 1960q2 thru 1990q4 Number of obs = 123
Wald chi2(2) = 310.64

Log likelihood = -135.3513 Prob > chi2 = 0.0000

OPG
D.wpi Coefficient std. err. z P>|z| [95% conf. interval]

wpi
_cons .7498197 .3340968 2.24 0.025 .0950019 1.404637

ARMA
ar

L1. .8742288 .0545435 16.03 0.000 .7673256 .981132

ma
L1. -.4120458 .1000284 -4.12 0.000 -.6080979 -.2159938

/sigma .7250436 .0368065 19.70 0.000 .6529042 .7971829

Note: The test of the variance against zero is one sided, and the two-sided
confidence interval is truncated at zero.

Examining the estimation results, we see that the AR(1) coefficient is 0.874, the MA(1) coefficient
is −0.412, and both are highly significant. The estimated standard deviation of the white-noise
disturbance ε is 0.725.

This model also could have been fit by typing

. arima D.wpi, ar(1) ma(1)

The D. placed in front of the dependent variable wpi is the Stata time-series operator for differencing.
Thus we would be modeling the first difference in WPI from the second quarter of 1960 through
the fourth quarter of 1990 because the first observation is lost because of differencing. This second
syntax allows a richer choice of models.

Example 2: ARIMA model with additive seasonal effects

After examining first differences of WPI, Enders chose a model of differences in the natural
logarithms to stabilize the variance in the differenced series. The raw data and first difference of the
logarithms are graphed below.

25

50

75

100

125

1960q1 1970q1 1980q1 1990q1
Quarterly date

US Wholesale Price Index

-.04

-.02

0

.02

.04

.06

.08

1960q1 1970q1 1980q1 1990q1
Quarterly date

US Wholesale Price Index -- difference of logs

On the basis of the autocorrelations, partial autocorrelations (see graphs below), and the results of
preliminary estimations, Enders identified an ARMA model in the log-differenced series.

. ac D.ln_wpi, ylabels(-.4(.2).6)

. pac D.ln_wpi, ylabels(-.4(.2).6)

In addition to an autoregressive term and an MA(1) term, an MA(4) term is included to account
for a remaining quarterly effect. Thus the model to be fit is

∆ ln(wpit) = β0 + ρ1{∆ ln(wpit−1)− β0}+ θ1εt−1 + θ4εt−4 + εt

We can fit this model with arima and Stata’s standard difference operator:

. arima D.ln_wpi, ar(1) ma(1 4)

(setting optimization to BHHH)
Iteration 0: Log likelihood = 382.67447
Iteration 1: Log likelihood = 384.80754
Iteration 2: Log likelihood = 384.84749
Iteration 3: Log likelihood = 385.39213
Iteration 4: Log likelihood = 385.40983
(switching optimization to BFGS)
Iteration 5: Log likelihood = 385.9021
Iteration 6: Log likelihood = 385.95646
Iteration 7: Log likelihood = 386.02979
Iteration 8: Log likelihood = 386.03326
Iteration 9: Log likelihood = 386.03354
Iteration 10: Log likelihood = 386.03357

ARIMA regression

Sample: 1960q2 thru 1990q4 Number of obs = 123
Wald chi2(3) = 333.60

Log likelihood = 386.0336 Prob > chi2 = 0.0000

OPG
D.ln_wpi Coefficient std. err. z P>|z| [95% conf. interval]

ln_wpi
_cons .0110493 .0048349 2.29 0.022 .0015731 .0205255

ARMA
ar

L1. .7806991 .0944946 8.26 0.000 .5954931 .965905

ma
L1. -.3990039 .1258753 -3.17 0.002 -.6457149 -.1522928
L4. .3090813 .1200945 2.57 0.010 .0737003 .5444622

/sigma .0104394 .0004702 22.20 0.000 .0095178 .0113609

Note: The test of the variance against zero is one sided, and the two-sided
confidence interval is truncated at zero.

In this final specification, the log-differenced series is still highly autocorrelated at a level of 0.781,
though innovations have a negative impact in the ensuing quarter (−0.399) and a positive seasonal
impact of 0.309 in the following year.

Technical note
In one way, the results differ from most of Stata’s estimation commands: the standard error of

the coefficients is reported as OPG std. err. The default standard errors and covariance matrix
for arima estimates are derived from the outer product of gradients (OPG). This is one of three
asymptotically equivalent methods of estimating the covariance matrix of the coefficients (only two of
which are usually tractable to derive). Discussions and derivations of all three estimates can be found
in Davidson and MacKinnon (1993), Greene (2018), and Hamilton (1994). Bollerslev, Engle, and
Nelson (1994) suggest that the OPG estimates are more numerically stable in time-series regressions
when the likelihood and its derivatives depend on recursive computations, which is certainly the case
for the Kalman filter. To date, we have found no numerical instabilities in either estimate of the
covariance matrix—subject to the stability and convergence of the overall model.

Most of Stata’s estimation commands provide covariance estimates derived from the Hessian of
the likelihood function. These alternate estimates can also be obtained from arima by specifying the
vce(oim) option.

Multiplicative seasonal ARIMA models

Many time series exhibit a periodic seasonal component, and a seasonal ARIMA model, often
abbreviated SARIMA, can then be used. For example, monthly sales data for air conditioners have a
strong seasonal component, with sales high in the summer months and low in the winter months.

In the previous example, we accounted for quarterly effects by fitting the model

(1− ρ1L){∆ ln(wpit)− β0} = (1 + θ1L+ θ4L
4)εt

This is an additive seasonal ARIMA model, in the sense that the first- and fourth-order MA terms work
additively: (1 + θ1L+ θ4L

4).

Another way to handle the quarterly effect would be to fit a multiplicative seasonal ARIMA model.
A multiplicative SARIMA model of order (1, 1, 1)× (0, 0, 1)4 for the ln(wpit) series is

(1− ρ1L){∆ ln(wpit)− β0} = (1 + θ1L)(1 + θ4,1L
4)εt

or, upon expanding terms,

∆ ln(wpit) = β0 + ρ1{∆ ln(wpit−1)− β0}+ θ1εt−1 + θ4,1εt−4 + θ1θ4,1εt−5 + εt (2)

In the notation (1, 1, 1)× (0, 0, 1)4, the (1, 1, 1) means that there is one nonseasonal autoregressive
term (1 − ρ1L) and one nonseasonal moving-average term (1 + θ1L) and that the time series is
first-differenced one time. The (0, 0, 1)4 indicates that there is no lag-4 seasonal autoregressive term,
that there is one lag-4 seasonal moving-average term (1 + θ4,1L

4), and that the series is seasonally
differenced zero times. This is known as a multiplicative SARIMA model because the nonseasonal
and seasonal factors work multiplicatively: (1 + θ1L)(1 + θ4,1L

4). Multiplying the terms imposes
nonlinear constraints on the parameters of the fifth-order lagged values; arima imposes these constraints
automatically.

To further clarify the notation, consider a (2, 1, 1)× (1, 1, 2)4 multiplicative SARIMA model:

(1− ρ1L− ρ2L2)(1− ρ4,1L4)∆∆4zt = (1 + θ1L)(1 + θ4,1L
4 + θ4,2L

8)εt (3)

where ∆ denotes the difference operator ∆yt = yt − yt−1 and ∆s denotes the lag-s seasonal
difference operator ∆syt = yt − yt−s. Expanding (3), we have

z̃t = ρ1z̃t−1 + ρ2z̃t−2 + ρ4,1z̃t−4 − ρ1ρ4,1z̃t−5 − ρ2ρ4,1z̃t−6
+ θ1εt−1 + θ4,1εt−4 + θ1θ4,1εt−5 + θ4,2εt−8 + θ1θ4,2εt−9 + εt

where
z̃t = ∆∆4zt = ∆(zt − zt−4) = zt − zt−1 − (zt−4 − zt−5)

and zt = yt − xtβ if regressors are included in the model, zt = yt − β0 if just a constant term is
included, and zt = yt otherwise.

More generally, a (p, d, q)× (P,D,Q)s multiplicative SARIMA model is

ρ(Lp)ρs(L
P )∆d∆D

s zt = θ(Lq)θs(L
Q)εt

where
ρs(L

P ) = (1− ρs,1Ls − ρs,2L2s − · · · − ρs,PLPs)
θs(L

Q) = (1 + θs,1L
s + θs,2L

2s + · · ·+ θs,QL
Qs)

ρ(Lp) and θ(Lq) were defined previously, ∆d means apply the ∆ operator d times, and similarly
for ∆D

s . Typically, d and D will be 0 or 1; and p, q, P , and Q will seldom be more than 2 or 3. s
will typically be 4 for quarterly data and 12 for monthly data. In fact, the model can be extended to
include both monthly and quarterly seasonal factors, as we explain below.

If a plot of the data suggests that the seasonal effect is proportional to the mean of the series, then
the seasonal effect is probably multiplicative and a multiplicative SARIMA model may be appropriate.
Box et al. (2016, sec. 9.3.1) suggest starting with a multiplicative SARIMA model with any data that
exhibit seasonal patterns and then exploring nonmultiplicative SARIMA models if the multiplicative
models do not fit the data well. On the other hand, Chatfield (2004, 14) suggests that taking the
logarithm of the series will make the seasonal effect additive, in which case an additive SARIMA
model as fit in the previous example would be appropriate. In short, the analyst should probably try
both additive and multiplicative SARIMA models to see which provides better fits and forecasts.

Unless diffuse is used, arima must create square matrices of dimension {max(p, q+1)}2, where
p and q are the maximum AR and MA lags, respectively; and the inclusion of long seasonal terms can
make this dimension rather large. For example, with monthly data, you might fit a (0, 1, 1)×(0, 1, 2)12
SARIMA model. The maximum MA lag is 2× 12 + 1 = 25, requiring a matrix with 262 = 676 rows
and columns.

Example 3: Multiplicative SARIMA model

One of the most common multiplicative SARIMA specifications is the (0, 1, 1)×(0, 1, 1)12 “airline”
model of Box et al. (2016, sec. 9.2). The dataset airline.dta contains monthly international airline
passenger data from January 1949 through December 1960. After first- and seasonally differencing
the data, we do not suspect the presence of a trend component, so we use the noconstant option
with arima:

. use https://www.stata-press.com/data/r18/air2
(TIMESLAB: Airline passengers)

. generate lnair = ln(air)

. arima lnair, arima(0,1,1) sarima(0,1,1,12) noconstant

(setting optimization to BHHH)
Iteration 0: Log likelihood = 223.8437
Iteration 1: Log likelihood = 239.80405

(output omitted )
Iteration 8: Log likelihood = 244.69651

ARIMA regression

Sample: 14 thru 144 Number of obs = 131
Wald chi2(2) = 84.53

Log likelihood = 244.6965 Prob > chi2 = 0.0000

OPG
DS12.lnair Coefficient std. err. z P>|z| [95% conf. interval]

ARMA
ma

L1. -.4018324 .0730307 -5.50 0.000 -.5449698 -.2586949

ARMA12
ma

L1. -.5569342 .0963129 -5.78 0.000 -.745704 -.3681644

/sigma .0367167 .0020132 18.24 0.000 .0327708 .0406625

Note: The test of the variance against zero is one sided, and the two-sided
confidence interval is truncated at zero.

Thus our model of the monthly number of international airline passengers is

∆∆12lnairt = −0.402εt−1 − 0.557εt−12 + 0.224εt−13 + εt

σ̂ = 0.037

In (2), for example, the coefficient on εt−13 is the product of the coefficients on the εt−1 and εt−12
terms (0.224 ≈ −0.402 × −0.557). arima labeled the dependent variable DS12.lnair to indicate
that it has applied the difference operator ∆ and the lag-12 seasonal difference operator ∆12 to
lnair; see [U] 11.4.4 Time-series varlists for more information.

We could have fit this model by typing
. arima DS12.lnair, ma(1) mma(1, 12) noconstant

For simple multiplicative models, using the sarima() option is easier, though this second syntax
allows us to incorporate more complicated seasonal terms.

The mar() and mma() options can be repeated, allowing us to control for multiple seasonal
patterns. For example, we may have monthly sales data that exhibit a quarterly pattern as businesses
purchase our product at the beginning of calendar quarters when new funds are budgeted, and our
product is purchased more frequently in a few months of the year than in most others, even after we
control for quarterly fluctuations. Thus we might choose to fit the model

(1−ρL)(1−ρ4,1L4)(1−ρ12,1L12)(∆∆4∆12salest−β0) = (1+θL)(1+θ4,1L
4)(1+θ12,1L

12)εt

Although this model looks rather complicated, estimating it using arima is straightforward:
. arima DS4S12.sales, ar(1) mar(1, 4) mar(1, 12) ma(1) mma(1, 4) mma(1, 12)

https://www.stata.com/manuals/u11.pdf#u11.4.4Time-seriesvarlists

If we instead wanted to include two lags in the lag-4 seasonal AR term and the first and third (but
not the second) term in the lag-12 seasonal MA term, we would type

. arima DS4S12.sales, ar(1) mar(1 2, 4) mar(1, 12) ma(1) mma(1, 4) mma(1 3, 12)

However, models with multiple seasonal terms can be difficult to fit. Usually, one seasonal factor
with just one or two AR or MA terms is adequate.

ARMAX models
Thus far all our examples have been pure ARIMA models in which the dependent variable was

modeled solely as a function of its past values and disturbances. Also, arima can fit ARMAX models,
which model the dependent variable in terms of a linear combination of independent variables, as
well as an ARMA disturbance process. The prais command (see [TS] prais), for example, allows
you to control for only AR(1) disturbances, whereas arima allows you to control for a much richer
dynamic error structure. arima allows for both nonseasonal and seasonal ARMA components in the
disturbances.

Example 4: ARMAX model

For a simple example of a model including covariates, we can estimate an update of Friedman and
Meiselman’s (1963) equation representing the quantity theory of money. They postulate a straight-
forward relationship between personal-consumption expenditures (consump) and the money supply
as measured by M2 (m2).

consumpt = β0 + β1m2t + µt

Friedman and Meiselman fit the model over a period ending in 1956; we will refit the model over
the period 1959q1 through 1981q4. We restrict our attention to the period prior to 1982 because the
Federal Reserve manipulated the money supply extensively in the later 1980s to control inflation, and
the relationship between consumption and the money supply becomes much more complex during
the later part of the decade.

To demonstrate arima, we will include both an autoregressive term and a moving-average term for
the disturbances in the model; the original estimates included neither. Thus we model the disturbance
of the structural equation as

µt = ρµt−1 + θεt−1 + εt

As per the original authors, the relationship is estimated on seasonally adjusted data, so there is no
need to include seasonal effects explicitly. Obtaining seasonally unadjusted data and simultaneously
modeling the structural and seasonal effects might be preferable.

We will restrict the estimation to the desired sample by using the tin() function in an if
expression; see [FN] Selecting time-span functions. By leaving the first argument of tin() blank,
we are including all available data through the second date (1981q4). We fit the model by typing

https://www.stata.com/manuals/tsprais.pdf#tsprais
https://www.stata.com/manuals/fnselectingtime-spanfunctions.pdf#fnSelectingtime-spanfunctionstin()
https://www.stata.com/manuals/fnselectingtime-spanfunctions.pdf#fnSelectingtime-spanfunctions

. use https://www.stata-press.com/data/r18/friedman2, clear

. arima consump m2 if tin(, 1981q4), ar(1) ma(1)

(setting optimization to BHHH)
Iteration 0: Log likelihood = -344.67575
Iteration 1: Log likelihood = -341.57248

(output omitted )
Iteration 10: Log likelihood = -340.50774

ARIMA regression

Sample: 1959q1 thru 1981q4 Number of obs = 92
Wald chi2(3) = 4394.80

Log likelihood = -340.5077 Prob > chi2 = 0.0000

OPG
consump Coefficient std. err. z P>|z| [95% conf. interval]

consump
m2 1.122029 .0363563 30.86 0.000 1.050772 1.193286

_cons -36.09872 56.56703 -0.64 0.523 -146.9681 74.77062

ARMA
ar

L1. .9348486 .0411323 22.73 0.000 .8542308 1.015467

ma
L1. .3090592 .0885883 3.49 0.000 .1354293 .4826891

/sigma 9.655308 .5635157 17.13 0.000 8.550837 10.75978

Note: The test of the variance against zero is one sided, and the two-sided
confidence interval is truncated at zero.

We find a relatively small money velocity with respect to consumption (1.122) over this period,
although consumption is only one facet of the income velocity. We also note a very large first-order
autocorrelation in the disturbances, as well as a statistically significant first-order moving average.

We might be concerned that our specification has led to disturbances that are heteroskedastic or
non-Gaussian. We refit the model by using the vce(robust) option.

. arima consump m2 if tin(, 1981q4), ar(1) ma(1) vce(robust)

(setting optimization to BHHH)
Iteration 0: Log pseudolikelihood = -344.67575
Iteration 1: Log pseudolikelihood = -341.57248

(output omitted )
Iteration 10: Log pseudolikelihood = -340.50774

ARIMA regression

Sample: 1959q1 thru 1981q4 Number of obs = 92
Wald chi2(3) = 1176.26

Log pseudolikelihood = -340.5077 Prob > chi2 = 0.0000

Semirobust
consump Coefficient std. err. z P>|z| [95% conf. interval]

consump
m2 1.122029 .0433302 25.89 0.000 1.037103 1.206954

_cons -36.09872 28.10478 -1.28 0.199 -91.18308 18.98564

ARMA
ar

L1. .9348486 .0493428 18.95 0.000 .8381385 1.031559

ma
L1. .3090592 .1605359 1.93 0.054 -.0055854 .6237038

/sigma 9.655308 1.082639 8.92 0.000 7.533375 11.77724

Note: The test of the variance against zero is one sided, and the two-sided
confidence interval is truncated at zero.

We note a substantial increase in the estimated standard errors, and our once clearly significant
moving-average term is now only marginally significant.

Dynamic forecasting

Another feature of the arima command is the ability to use predict afterward to make dynamic
forecasts. Suppose that we wish to fit the regression model

yt = β0 + β1xt + ρyt−1 + εt

by using a sample of data from t = 1 . . . T and make forecasts beginning at time f .

If we use regress or prais to fit the model, then we can use predict to make one-step-ahead
forecasts. That is, predict will compute

ŷf = β̂0 + β̂1xf + ρ̂yf−1

Most importantly, here predict will use the actual value of y at period f − 1 in computing the
forecast for time f . Thus, if we use regress or prais, we cannot make forecasts for any periods
beyond f = T + 1 unless we have observed values for y for those periods.

If we instead fit our model with arima, then predict can produce dynamic forecasts by using
the Kalman filter. If we use the dynamic(f) option, then for period f predict will compute

ŷf = β̂0 + β̂1xf + ρ̂yf−1

by using the observed value of yf−1 just as predict after regress or prais. However, for period
f + 1 predict newvar, dynamic(f) will compute

ŷf+1 = β̂0 + β̂1xf+1 + ρ̂ŷf

using the predicted value of yf instead of the observed value. Similarly, the period f + 2 forecast
will be

ŷf+2 = β̂0 + β̂1xf+2 + ρ̂ŷf+1

Of course, because our model includes the regressor xt, we can make forecasts only through periods
for which we have observations on xt. However, for pure ARIMA models, we can compute dynamic
forecasts as far beyond the final period of our dataset as desired.

For more information on predict after arima, see [TS] arima postestimation.

Video example

Introduction to ARMA/ARIMA models

Stored results
arima stores the following in e():

Scalars
e(N) number of observations
e(N gaps) number of gaps
e(k) number of parameters
e(k eq) number of equations in e(b)
e(k eq model) number of equations in overall model test
e(k dv) number of dependent variables
e(k1) number of variables in first equation
e(df m) model degrees of freedom
e(ll) log likelihood
e(sigma) sigma
e(chi2) χ2

e(p) p-value for model test
e(tmin) minimum time
e(tmax) maximum time
e(ar max) maximum AR lag
e(ma max) maximum MA lag
e(rank) rank of e(V)
e(ic) number of iterations
e(rc) return code
e(converged) 1 if converged, 0 otherwise

https://www.stata.com/manuals/tsarimapostestimation.pdf#tsarimapostestimation
https://www.youtube.com/watch?v=8xt4q7KHfBs

Macros
e(cmd) arima
e(cmdline) command as typed
e(depvar) name of dependent variable
e(covariates) list of covariates
e(eqnames) names of equations
e(wtype) weight type
e(wexp) weight expression
e(title) title in estimation output
e(tmins) formatted minimum time
e(tmaxs) formatted maximum time
e(chi2type) Wald; type of model χ2 test
e(vce) vcetype specified in vce()
e(vcetype) title used to label Std. err.
e(ma) lags for moving-average terms
e(ar) lags for autoregressive terms
e(mari) multiplicative AR terms and lag i=1... (# seasonal AR terms)
e(mmai) multiplicative MA terms and lag i=1... (# seasonal MA terms)
e(seasons) seasonal lags in model
e(opt) type of optimization
e(ml method) type of ml method
e(user) name of likelihood-evaluator program
e(technique) maximization technique
e(tech steps) number of iterations performed before switching techniques
e(properties) b V
e(estat cmd) program used to implement estat
e(predict) program used to implement predict
e(marginsok) predictions allowed by margins
e(marginsnotok) predictions disallowed by margins

Matrices
e(b) coefficient vector
e(Cns) constraints matrix
e(ilog) iteration log (up to 20 iterations)
e(gradient) gradient vector
e(V) variance–covariance matrix of the estimators
e(V modelbased) model-based variance

Functions
e(sample) marks estimation sample

In addition to the above, the following is stored in r():

Matrices
r(table) matrix containing the coefficients with their standard errors, test statistics, p-values,

and confidence intervals

Note that results stored in r() are updated when the command is replayed and will be replaced when
any r-class command is run after the estimation command.

Methods and formulas
Estimation is by maximum likelihood using the Kalman filter via the prediction error decomposition;

see Hamilton (1994), Gouriéroux and Monfort (1997), or, in particular, Harvey (1989). Any of these
sources will serve as excellent background for the fitting of these models with the state-space form;
each source also provides considerable detail on the method outlined below.

Methods and formulas are presented under the following headings:

ARIMA model
Kalman filter equations
Kalman filter or state-space representation of the ARIMA model
Kalman filter recursions
Kalman filter initial conditions
Likelihood from prediction error decomposition
Missing data

ARIMA model

The model to be fit is

yt = xtβ + µt

µt =

p∑
i=1

ρiµt−i +

q∑
j=1

θjεt−j + εt

which can be written as the single equation

yt = xtβ +

p∑
i=1

ρi(yt−i − xt−iβ) +

q∑
j=1

θjεt−j + εt

Some of the ρs and θs may be constrained to zero or, for multiplicative seasonal models, the products
of other parameters.

Kalman filter equations

We will roughly follow Hamilton’s (1994) notation and write the Kalman filter

ξt = Fξt−1 + vt (state equation)

yt = A′xt + H′ξt + wt (observation equation)

and (
vt
wt

)
∼ N

{
0,

(
Q 0
0 R

)}
We maintain the standard Kalman filter matrix and vector notation, although for univariate models
yt, wt, and R are scalars.

Kalman filter or state-space representation of the ARIMA model

A univariate ARIMA model can be cast in state-space form by defining the Kalman filter matrices
as follows (see Hamilton [1994], or Gouriéroux and Monfort [1997], for details):

F =


ρ1 ρ2 . . . ρp−1 ρp
1 0 . . . 0 0
0 1 . . . 0 0
0 0 . . . 1 0



vt =


εt−1

0
. . .
. . .
. . .
0


A′ = β

H′ = [ 1 θ1 θ2 . . . θq ]

wt = 0

The Kalman filter representation does not require the moving-average terms to be invertible.

Kalman filter recursions
To demonstrate how missing data are handled, the updating recursions for the Kalman filter will

be left in two steps. Writing the updating equations as one step using the gain matrix K is common.
We will provide the updating equations with little justification; see the sources listed above for details.

As a linear combination of a vector of random variables, the state ξt can be updated to its expected
value on the basis of the prior state as

ξt|t−1 = Fξt−1 + vt−1 (4)

This state is a quadratic form that has the covariance matrix

Pt|t−1 = FPt−1F
′ + Q (5)

The estimator of yt is
ŷt|t−1 = xtβ + H′ξt|t−1

which implies an innovation or prediction error

ι̂t = yt − ŷt|t−1

This value or vector has mean squared error (MSE)

Mt = H′Pt|t−1H + R

Now the expected value of ξt conditional on a realization of yt is

ξt = ξt|t−1 + Pt|t−1HM−1t ι̂t (6)

with MSE
Pt = Pt|t−1 −Pt|t−1HM−1t H′Pt|t−1 (7)

This expression gives the full set of Kalman filter recursions.



22 arima — ARIMA, ARMAX, and other dynamic regression models

Kalman filter initial conditions
When the series is stationary, conditional on xtβ, the initial conditions for the filter can be

considered a random draw from the stationary distribution of the state equation. The initial values of
the state and the state MSE are the expected values from this stationary distribution. For an ARIMA
model, these can be written as

ξ1|0 = 0

and
vec(P1|0) = (Ir2 − F⊗ F)−1vec(Q)

where vec() is an operator representing the column matrix resulting from stacking each successive
column of the target matrix.

If the series is not stationary, the initial state conditions do not constitute a random draw from a
stationary distribution, and some other values must be chosen. Hamilton (1994) suggests that they be
chosen based on prior expectations, whereas Harvey suggests a diffuse and improper prior having a
state vector of 0 and an infinite variance. This method corresponds to P1|0 with diagonal elements of
∞. Stata allows either approach to be taken for nonstationary series—initial priors may be specified
with state0() and p0(), and a diffuse prior may be specified with diffuse.

Likelihood from prediction error decomposition

Given the outputs from the Kalman filter recursions and assuming that the state and observation
vectors are Gaussian, the likelihood for the state-space model follows directly from the resulting
multivariate normal in the predicted innovations. The log likelihood for observation t is

lnLt = −1

2

{
ln(2π) + ln(|Mt|)− ι̂′tM

−1
t ι̂t

}
This command supports the Huber/White/sandwich estimator of the variance using vce(robust).

See [P] robust, particularly Maximum likelihood estimators and Methods and formulas.

Missing data

Missing data, whether a missing dependent variable yt, one or more missing covariates xt, or
completely missing observations, are handled by continuing the state-updating equations without any
contribution from the data; see Harvey (1989 and 1993). That is, (4) and (5) are iterated for every
missing observation, whereas (6) and (7) are ignored. Thus, for observations with missing data,
ξt = ξt|t−1 and Pt = Pt|t−1. Without any information from the sample, this effectively assumes
that the prediction error for the missing observations is 0. Other methods of handling missing data
on the basis of the EM algorithm have been suggested, for example, Shumway (1984, 1988).

https://www.stata.com/manuals/p_robust.pdf#p_robust
https://www.stata.com/manuals/p_robust.pdf#p_robustRemarksandexamplesMaximumlikelihoodestimators
https://www.stata.com/manuals/p_robust.pdf#p_robustMethodsandformulas

George Edward Pelham Box (1919–2013) was born in Kent, England, and earned degrees
in statistics at the University of London. After work in the chemical industry, he taught and
researched at Princeton and the University of Wisconsin. His many major contributions to statistics
include papers and books in Bayesian inference, robustness (a term he introduced to statistics),
modeling strategy, experimental design and response surfaces, time-series analysis, distribution
theory, transformations, and nonlinear estimation.

Gwilym Meirion Jenkins (1933–1982) was a British mathematician and statistician who spent
his career in industry and academia, working for extended periods at Imperial College London
and the University of Lancaster before running his own company. His interests were centered on
time series and he collaborated with G. E. P. Box on what are often called Box–Jenkins models.
The last years of Jenkins’s life were marked by a slowly losing battle against Hodgkin’s disease.� �

